
EJERCICIO 4 – EEPROM- 
 
 
CONEXIONES DE LA PLACA BASICA DE APENDIZAJE: 

 
 Para este ejercicio necesitamos la placa básica de aprendizaje, en ella tenemos 
conectado a las líneas RA0..RA3, 4 diodos led y a las líneas RB5, RB4 y RB0, tres 
pulsadores de la siguiente manera: 
 
 
         RAx           RBx 
   330 oh.        10 oh. 
         100 nF 
  
 
  
EJERCICIO 4-0: 

 
 El objetivo de este ejercicio es construir un contador con el PORTA cuyo valor 
binario se  visualizará en los leds. Para incrementar el valor del contador pulsaremos la 
tecla conectada a RB0. Este valor se almacenará en la eeprom de tal manera que si se 
apaga y se vuelve a encender el integrado, la cuenta se conservará. 
 

La EEPROM del  PIC16F84: 

 
 Para manejar la EEprom el PIC tiene 4 registros, estos son: 
 

• EEADR(09h): En este registro almacenamos la dirección del byte que queremos 
leer o escribir. Para el PIC16F84 el rango de direcciones va desde 00h-3Fh, que 
son 64 bytes. 

• EEDATA(08h): En este registro se almacena el dato a grabar o leido. 
• EECON2(89h): Este es un registro no implementado y se utiliza para meter una 

clave para la escritura. 
• EECON1(88h): Este es el registro que controla la lectura y escritura en la 

eeprom y posee la siguiente información: 
 

Bit 7 Bit 6 Bit 5 Bit  4 Bit 3 Bit 2 Bit 1 Bit 0 
- - - EEIF WRERR WREN WR RD 

 
o EEIF: Este es un banderín que se pone a 1 si la interrupción generada es 

de final de escritura en eeprom. 
o WRERR: Si este bit esta a 1 implica que ha habido un error en la 

grabación del byte, por efecto de un reset de WDT o MCLR. 
o WREN: Si lo ponemos a 1 se permite la grabación en la EEprom.    
o WR: Si lo ponemos a 1 iniciamos el proceso de grabación del byte, 

cuando termina este se pone a 0 solo. 
o RD: Si lo ponemos a 1 iniciamos el proceso de lectura de un byte, 

cuando termina este se pone a 0. 
 
 



Para leer un byte, por ejemplo, de la posición 00h, haríamos lo siguiente: 
 

1. Cargaríamos EEADR con 00h. 
2. Pondríamos a 1 el bit RD del registro EECON1. 
3. Esperamos que sea 0. 
4. El registro EEDATA contiene el dato leído. 

 
Para grabar un byte, por ejemplo, de la posición 00h con el valor 05h, haríamos 

lo siguiente: 
 

5. Cargaríamos EEADR con 00h y EEDATA con 05h. 
6. Pondríamos a 1 el bit WREN del registro EECON1. 
7. Grabamos 55h en el registro EECON2 
8. Grabamos AAh en el registro EECON2 
9. Ponemos a 1 el bit WR del registro EECON1  
10. Esperamos que sea 0. 

 
ATENCIÓN: Durante el proceso de escritura es imprescindible que no se 

produzcan interrupciones, por lo que deberemos inhibirlas antes de entrar en la 
grabación, luego podemos activarlas. 

 
 
Para estudiar todo esto, utilizaremos el fichero Ejer4-0.asm cuyo código es el 

siguiente: 
 
  Include p16f84.inc 
 
  org H'0000 
  goto Inicio 
 
  org H'0004   ;VECTOR DE INTERRUPCION 
  btfsc INTCON,INTF  ;Comprueba si la int. es de RB0/INT 
  goto IntExt 
  retfie 
IntExt: incf PORTA 
  movlw H'00 
  movwf EEADR 
  movf PORTA,w 
  movwf EEDATA 
  call Escdato 
  bcf INTCON,INTF 
  retfie 
 
 
Inicio: Bsf STATUS,RP0  ;Activa la pagina 1 
  Movlw b'11110000  ;RA0..RA4 como salida. 
  Movwf TRISA  
  Movlw b'11111111  ;Todo el puerto B como entrada. 
  Movwf TRISB 
  bcf OPTION_REG,NOT_RBPU ;Pull-up en puerto B 
  bcf OPTION_REG,INTEDG ;Flanco de bajada en interrupcion. 
  Bcf STATUS,RP0  ;Activa la pagina 0 
 
  movlw H'00   ;Lee posicion 0 de la EEPROM. 
  movwf EEADR 
  call leeDato 



  Movwf PORTA   ;Transfiere W al puerto RA. 
 
  bsf INTCON,GIE  ;Activa todas las interrupciones. 
  bsf INTCON,INTE  ;Activa la interrupcion RA0/INT. 
 
Fin:  Goto fin   ;Bucle de parada. 
 
;PROCEDIMIENTO DE LECTURA DE LA EEPROM. 
 
LeeDato: bsf STATUS,RP0  ;EEADR posee la dirección a leer 
  bsf EECON1,RD 
  nop 
BuleeDato: btfsc EECON1,RD 
  goto buleeDato 
  bcf STATUS,RP0 
  movf EEDATA,w 
  return 
 
;PROCEDIMIENTO DE ESCRITURA DE LA EEPROM. 
   
EscDato: bsf STATUS,RP0  ;Direccion en EEADR,dato en EEDATA.  
  bsf EECON1,WREN 
  movlw H'55 
  movwf EECON2 
  movlw H'AA 
  movwf EECON2 
  bsf EECON1,WR 
  nop 
BuEscDato: btfsc EECON1,WR 
  goto buEscDato 
  bcf STATUS,RP0 
  return 
 
  End 
 
 

Pasemos a analizar el programa: 
 
 Este ejemplo es el mismo del ejer3-0.asm, la diferencia esta en dos cuestiones; la 
primera es que en vez de inicializar el PORTA con 0, lo inicializamos con el valor 
almacenado en la posición 00h de la EEPROM y la segunda es que cada vez que 
incrementamos la cuenta, el nuevo valor se almacena en la posición 00h de la 
EEPROM. Observa que como la grabación del byte la efectuamos dentro de la 
interrupción no necesitamos inhibir las interrupciones, ya que estas ya están inhibidas. 
  
 El proceso de lectura y grabación lo realizan las dos subrutinas EscDato y 
LeeDato, estas siguen la secuencia del proceso de lectura y grabación explicada 
anteriormente. 

